

## metapost

# Aligning METAPOST graphs in CONTEXT combinations

## Introduction

For scientific plotting I like to use the Graph package by John Hobby within CONTEXT and when I have two or more separate graphs made I combine them into one figure with one figure caption. Combining is easy but aligning the graphs in a pleasing way required a trick.

### Attempt 1

To demonstrate the problem I prepared two graphs plotting blood pressure and heart interval versus time (of course, it could be any signal) and added the result of spot measurements (BRS) to the graph. The vertical scales differ for each signal and these scales must be indicated in the graph. Each graph in isolation looks beautiful and a particularly nice feature is its seemingly endless scalability. Blowing the graph up from a design size of 5 cm width to 50 cm width or bigger is no problem.

CONTEXT then performs the combination of the graphs in one column aligned vertically using the `\startcombination[1*2]` command. The METAPOST pictures have been given an invisible outer box size, the so-called “bounding box” and this is a narrow fit to the total spaced used. METAPOST doesn’t care about centering the actual graph area in the bounding box. The graph areas, therefore, are not aligned vertically. See Fig. 1 left two panels.

### Attempt 2

Adding the vertical scale values to the graphs, that differ in width, causes the miss-alignment. The plot areas are exactly 5 cm but the lettering sticks out on either side of it. This unbalancing effect could be annihilated by adding invisible texts of equal width, sticking out left and right of the graph area by an equal amount. An `\hbox` command comes to mind. An `\hbox` can be given a horizontal width. In CONTEXT a `\framed` command could also be used. Don’t try it, however, they don’t work unless they are filled with something that prints (thus not spaces) or as long as the empty `\framed` has a frame that’s not made invisible. Clearly, METAPOST refuses to adapt its bounding box to invisible matter. We, on the other hand, refuse to have extraneous matter printed.

Here is a solution that works, nevertheless. We let METAPOST draw a `fullsquare xyscaled(10cm,8cm)` centered at the midpoint of the graph area. This makes both graphs of equal width and height and alignment in a `\startcombination` is perfect. This is shown in Fig. 1 right two panels. The `fullsquare` is opaque. Anything plotted before it is drawn disappears behind it, anything plotted later is visible. It must therefore be the first plot command in the picture. In addition, it must be given a color and the color might be considered undesirable. The solution to that is to plot it with `color .9999white`. For all practical purposes this is as white as white paper is.

Here is the program for two curves within one graph:

```

\startreusableMPgraphic{outlier}
  picture Panel;           % .9999white for a white panel
  Panel:=image(fill fullsquare xyscaled(10cm,8cm) withcolor .9white);
  draw begingraph(5cm,5cm); % the plot area is 5 x 5 cm
  setcoords(linear,linear); % always set coordinates, or else
  setrange(210,100,230,125); % then set range
  gdraw (220,112.5) plot Panel; % center horizontal & vertical
  glabel.top(btex ibi (ms) etex,231,126);
  glabel.bot(btex time (s) etex,220,96);
  gdraw "sy13lb.ups" withpen pencircle scaled 1.5bp; % data plot
  autogrid(,);           % always an autogrid, here empty
  setcoords(linear,linear); % hidden effect: resets scaling
  setrange(210,500,230,1500); % note a different vertical scale
  gdraw "in13lb.ups" withpen pencircle scaled .7bp; % data plot
  autogrid(otick.bot,grid.rt); % draw horizontal & vertical scales
  endgraph;
\stopreusableMPgraphic

```

And this is how two of them are used in a CONTEXT combination.


```

\placefigure[middle][fig:outlierb013]
  {Both left panels...
  }
  {\startcombination[1*2]
    {\reuseMPgraphic{outlier1}} {}
    {\reuseMPgraphic{outlier2}} {}
  \stopcombination}

```

## Postscript

A true hacker would have thought of a solution in less time than it cost me. The manual that comes with Hobby's Graph package is written for experts, not for the common user. Graph's output is so accurate and scalable that one wants to use it anyway but experimentation is time consuming and when one must finish a production on time and within budget the level of experimentation required is bedeviling. The solution for that is to post a question with the CONTEXT user group.



**Figure 1** Both left panels on white background are clearly shifted with respect to each other as indicated by the dashed lines that do not meet. Right panels on a gray background show a correctly aligned system of diagrams. In all diagrams: the thick line represents systolic pressure variations in mmHg, over a 20 s observation window in a person (use 'sys' vertical scale), the thin line similarly is the time interval between heart beats in ms (use 'ibi' vertical scale), the crosses represent estimates of the sensitivity of blood pressure control by the bodies baroreflex expressed in ms/mmHg (use brs vertical scale) derived from the combined blood pressure and beat interval variability data which clearly move up and down in synchronism.